TD L3 Algèbre et géométrie - Feuille 2 ex. 3 avec Sagemath

F-X. Dehon - 3 oct. 2020 - dehon[@]unice.fr

Enoncé :

> **3.** Soit $E = \mathbb{R}_2[X]$ et $\langle P, Q\rangle = \int_{-1}^{+1} P(t)Q(t)dt$.
>
> **3.a.** Montrer que $(E, \langle -, -\rangle)$ est un espace euclidien de dimension 3.
> **3.b.** En appliquant le procédé de Gram-Schmidt à $(1, X, X^2)$ trouver une base orthonormée (L_0, L_1, L_2) de $(E, \langle -, -\rangle)$.
> **3.c.** Montrer que $P \mapsto u(P) = (1 - X^2)P'(X) + (2X + 1)P(X)$ définit un endomorphisme $u : E \to E$. Ecrire u dans la base canonique $(1, X, X^2)$.
> **3.d.** Ecrire u dans la base (L_0, L_1, L_2). En déduire la valeur de l'intégrale $\int_{-1}^{1}(u(P))^2 dt$ en fonction des coefficients (a, b, c) de $P(X) = aL_0 + bL_1 + cL_2$.

Les éléments de $E = \mathbb{R}_2[X]$ dans l'énoncé sont ici déclarés comme des fonctions de la variable x sans précision sur son type.

In [1]:
```
a=1/sqrt(2);
print a.parent(),",",a.category()
print a,AA(a),RR(a)
```

```
Symbolic Ring , Category of elements of Symbolic Ring
1/2*sqrt(2) 0.7071067811865475? 0.707106781186548
```

Plusieurs façons de déclarer une fonction, pas strictement équivalentes :

In [2]:
```
a=1/sqrt(2)
P0(x)=x+a
Q0=lambda x:x+a
def R0(x):return(x+a)
print P0,",",Q0,",",R0
print P0==Q0,bool(P0(x)==Q0(x))
print P0==R0,bool(P0(x)==R0(x))

print P0.derivative()
try:print Q0.derivative()
except Exception as e: print "Q0.derivative() :",e
print Q0(x).derivative(x)
```

```
x |--> x + 1/2*sqrt(2) , <function <lambda> at 0x7fee6856d5f0> , <function R0 at 0x7fee6
856d668>
False True
False True
x |--> 1
Q0.derivative() : 'function' object has no attribute 'derivative'
1
```

Définition de la forme $\langle -, -\rangle$ nommée B ci-dessous, de l'application u, coordonnées et matrices

In [3]:
```
def B(P,Q):return(integrate(P(x)*Q(x),x,-1,1))
```

In [4]:
```
print B(lambda x:x, lambda x:x^3)
```

```
2/5
```

In [5]:
```
A=matrix(3,3,lambda i,j:B(lambda x:x^i,lambda x:x^j));pretty_print(A)
```

Out[5]:

$$\begin{pmatrix} 2 & 0 & \frac{2}{3} \\ 0 & \frac{2}{3} & 0 \\ \frac{2}{3} & 0 & \frac{2}{5} \end{pmatrix}$$

In [6]:
```
var('a b c')
X=matrix([a,b,c]).transpose();pretty_print(X)
print (X.transpose()*A*X)[0,0].expand()
```

Out[6]:

$$\begin{pmatrix} a \\ b \\ c \end{pmatrix}$$

```
2*a^2 + 2/3*b^2 + 4/3*a*c + 2/5*c^2
```

In [7]:
```
P(x)=a+b*x+c*x^2
print B(P,P).expand()
```

```
2*a^2 + 2/3*b^2 + 4/3*a*c + 2/5*c^2
```

In [8]:
```
def u(P):return((1-x^2)*(P.derivative(x))+(2*x+1)*P(x))
```

In [9]:
```
print u(P)(x).expand()
```

```
b*x^2 + c*x^2 + 2*a*x + b*x + 2*c*x + a + b
```

In [10]:
```
print B(u(P),u(P)).expand()
```

```
14/3*a^2 + 8*a*b + 22/5*b^2 + 20/3*a*c + 24/5*b*c + 46/15*c^2
```

In [11]:
```
def coef(P):return([P(0),P.derivative(x)(0),P.derivative(x,2)(0)/2])
```

In [12]:
```
print coef(P)
```

```
[a, b, c]
```

In [13]:
```
P1(x)=1;P2(x)=x;P3(x)=x^2
Base=[P1,P2,P3]
Mu=matrix([coef(u(e)) for e in Base]).transpose();pretty_print(Mu)
```

Out[13]:

$$\begin{pmatrix} 1 & 1 & 0 \\ 2 & 1 & 2 \\ 0 & 1 & 1 \end{pmatrix}$$

In [14]:
```
print (X.transpose()*Mu.transpose()*A*Mu*X)[0,0].expand()
```

```
14/3*a^2 + 8*a*b + 22/5*b^2 + 20/3*a*c + 24/5*b*c + 46/15*c^2
```

Orthonormalisation de Gram-Schmidt de $(P1, P2, P3)$, coordonnées dans la nouvelle base

In [15]:
```
def N(P):return(P/sqrt(B(P,P)))
def proj(P,base):return(sum(B(P,L)*L for L in base))
```

In [16]:
```
L1=N(P1);print L1(x)
```

```
1/2*sqrt(2)
```

In [17]:
```
L2p=P2-proj(P2,[L1])
L2=N(P2);print L2(x)
```

```
3/2*sqrt(2/3)*x
```

In [18]:
```
L3p=P3-proj(P3,[L1,L2])
L3=N(L3p);print L3(x)
```

```
5/4*sqrt(2/5)*(3*x^2 - 1)
```

In [19]:
```
def coefL(P):return([B(P,L) for L in [L1,L2,L3]])
```

In [20]:
```
MLu=matrix([coefL(u(L)) for L in [L1,L2,L3]]).transpose()
pretty_print(MLu)
```

Out[20]:

$$\begin{pmatrix} 1 & \frac{4}{3}\sqrt{3} & 0 \\ \frac{2}{3}\sqrt{3} & 1 & \frac{2}{3}\sqrt{5}\sqrt{3} \\ 0 & \frac{2}{15}\sqrt{5}\sqrt{3} & 1 \end{pmatrix}$$

In [21]:
```
baseL=[L1,L2,L3]
pretty_print(matrix(3,3,lambda i,j:B(baseL[i],baseL[j])))
```

Out[21]:

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

In [22]:
```
P=a*L1+b*L2+c*L3
print B(u(P),u(P)).expand()
```

```
8/5*sqrt(5)*sqrt(3)*b*c + 4*sqrt(3)*a*b + 8/3*sqrt(5)*a*c + 7/3*a^2 + 33/5*b^2 + 23/3*c^
2
```

In [23]:
```
X=matrix([a,b,c]).transpose()
print (X.transpose()*MLu.transpose()*MLu*X)[0,0].expand()
```

```
8/5*sqrt(5)*sqrt(3)*b*c + 4*sqrt(3)*a*b + 8/3*sqrt(5)*a*c + 7/3*a^2 + 33/5*b^2 + 23/3*c^
2
```

Autre définition de E dans Sagemath

In [24]:
```
print sqrt(2).parent()
print SR,",",AA
```

```
Symbolic Ring
Symbolic Ring , Algebraic Real Field
```

In [25]:
```
S.<X>=SR[] #AA[]
print 1.parent(),",",S(1).parent()
print 1==S(1)
try:print 1.derivative()
except Exception as e:print(e)
print S(1).derivative()
P=1+X+X^2;print integrate(P(x),x,-1,1)
print B(P,P)
```

```
Integer Ring , Univariate Polynomial Ring in X over Symbolic Ring
1 == 1
'sage.rings.integer.Integer' object has no attribute 'derivative'
0
8/3
22/5
```

In [26]:
```
def u(P):return((1-X^2)*(P.derivative())+(2*X+1)*P)
print u(X^2)
v(P)=(1-X^2)*(P.derivative())+(2*X+1)*P
print v(X^2),"bizarre !!"
w(P)=derivative(P);print w(X^3);print derivative(X^3)
```

```
X^2 + 2*X
2*X^3 + 1 bizarre !!
1
3*X^2
```

Scholies